

## T.Pajdla: Inverse Kinematics of a 6-DOF Manipulator - 2

- [1] D.Manocha, J.F.Canny. Efficient Inverse Kinematics for General 6R Manipulators. IEEE Trans. on Robotics and Automation, 10(5), pp. 648-657, Oct. 2004  
[2] M. Raghavan, B. Roth. Kinematic Analysis of the 6R Manipulator of General Geometry. Int. Symposium on Robotic Research. pp. 264-269, Tokyo 1990

General Mechanism - Explanation  
2 Nov 2009

### — Packages & settings

```
> restart:
with(ListTools):
with(LinearAlgebra):
with(PolynomialTools):
with(combinat, choose):
with(Groebner):
with(MatrixPolynomialAlgebra):
interface(rtablesize=24):
interface(warnlevel=0):
Digits:=30:
eps:=1e-6:
```

### — DH-Kinematics functions

[ Joint transformations:

```
> # Two one-parametric motions transformatin in DH-convention(phi, theta, a,d)
indexed by i
# c = cos(phi), s = sin(phi), lambda = cos(alpha), mu = sin(alpha)
dhTs := proc(i)
local M1, M2;
M1:=Matrix(4,4,[[+cat(`c`,i),-cat(`s`,i),0,
0],
[ +cat(`s`,i),+cat(`c`,i),0,
0],
[ 0, 0,1,cat(`d`,i)],
[ 0, 0, 0,0, 1]]);
M2:=Matrix(4,4,[[1, 0, 0,cat(`a`,i)],
[ 0,+cat(`lambda`,i),-cat(`mu`,i), 0],
[ 0,+cat(`mu`,i),+cat(`lambda`,i), 0],
[ 0, 0, 0, 1]]);
[M1,M2];
end proc:
#
# Inverse of the DH-convention for one-aprametric DH rigid motion transformations
dhInvs := proc(M)
local M1, M2;
M1 := M[1];
M2 := M[2];
[simplify(MatrixInverse(M2),{M2[3,2]^2+M2[3,3]^2=1}),
simplify(MatrixInverse(M1),{M1[1,1]^2+M1[2,1]^2=1})];
end proc:
#
# Rigid motion transformatin in DH-convention(phi, theta, a,d) indexed by i
# c = cos(phi), s = sin(phi), P = cos(alpha), R = sin(alpha)
dhT := proc(i)
local M;
M:=dhTs(i);
M[1].M[2];
end proc:
#
# Inverse of the DH-convention rigid motion transformation
dhInv := proc(M)
simplify(MatrixInverse(M),{M[1,1]^2+M[2,1]^2=1,M[3,2]^2+M[3,3]^2=1});
end proc:
```

```

#
# Simplify using trigonometric indentities  $c^2+s^2=1$  &  $\lambda^2+\mu^2=1$ 
dhSimpl := proc(M,i)
    simplify(M,{cat(`c`,i)^2+cat(`s`,i)^2=1,cat(`lambda`,i)^2+cat(`mu`,i)^2=1});
end proc:
#
## Direct Kinematic Task
#
dhDKT := proc(p)
    subs(p,dhT(1).dhT(2).dhT(3).dhT(4).dhT(5).dhT(6));
end proc:
#
# Simplify using Rotation matrin in Mh
MhSimpl := proc(M)
    simplify(
        simplify(
            simplify(
                simplify(M,
                    {lx^2+ly^2+lz^2=1,mx^2+my^2+mz^2=1,nx^2+ny^2+nz^2=1}),
                    {lx*mx+ly*my+lz*mz=0,lx*nx+ly*ny+lz*nz=0,mx*nx+my*ny+mz*nz=0}),
                    {lx^2+mx^2+nx^2=1,ly^2+my^2+ny^2=1,lz^2+mz^2+nz^2=1}),
                    {lx*ly+mx*my+nx*ny=0,lx*lz+mx*mz+nx*nz=0,lz*ly+mz*my+nz*ny=0});
end proc:
#
# Simplify a general motion matrix using rotation matrix identities in columns
rcSimp := proc(M,R)
    simplify(
        simplify(
            simplify(
                simplify(
                    simplify(
                        simplify(M,{R[1,1]*R[1,1]+R[2,1]*R[2,1]+R[3,1]*R[3,1]=1}),
                        {R[1,1]*R[1,2]+R[2,1]*R[2,2]+R[3,1]*R[3,2]=0}),
                        {R[1,1]*R[1,3]+R[2,1]*R[2,3]+R[3,1]*R[3,3]=0}),
                        {R[1,2]*R[1,2]+R[2,2]*R[2,2]+R[3,2]*R[3,2]=1}),
                        {R[1,2]*R[1,3]+R[2,2]*R[2,3]+R[3,2]*R[3,3]=0}),
                        {R[1,3]*R[1,3]+R[2,3]*R[2,3]+R[3,3]*R[3,3]=1});
end proc:
#
# Simplify a general motion matrix using rotation matrix identities in rows
rrSimp := proc(M,R)
    simplify(
        simplify(
            simplify(
                simplify(
                    simplify(
                        simplify(M,{R[1,1]*R[1,1]+R[1,2]*R[1,2]+R[1,3]*R[1,3]=1}),
                        {R[1,1]*R[2,1]+R[1,2]*R[2,2]+R[1,3]*R[2,3]=0}),
                        {R[1,1]*R[3,1]+R[1,2]*R[3,2]+R[1,3]*R[3,3]=0}),
                        {R[2,1]*R[2,1]+R[2,2]*R[2,2]+R[2,3]*R[2,3]=1}),
                        {R[2,1]*R[3,1]+R[2,2]*R[3,2]+R[2,3]*R[3,3]=0}),
                        {R[3,1]*R[3,1]+R[3,2]*R[3,2]+R[3,3]*R[3,3]=1});
end proc:
#
# Matrix representation of a set of polynomials
PolyCoeffMatrix:=proc(S,m,Ord:={ShortTermOrder, TermOrder})
local A,v,i,j,k,c,q;
    A:=Matrix(nops(S),nops(m),storage=sparse);
    v:=indets(m);
    for i from 1 to nops(S) do
        c:=[coeffs(expand(S[i]),v,'q')];
        q:=[q];
    end do;
end proc:

```

```

                for j from 1 to nops(m) do
                    for k from 1 to nops(q) do
                        if (m[j]=q[k]) then A[i,j]:=c[k] end if
                    end do
                end do
            end do;
            Matrix(A);
end proc:
#
## Cartesian product of a two lists
#
LxL:=proc(X::list,Y::list)
    Flatten(map(x->(map(y->Flatten([x,y]),Y)),X),1);
end proc:
#
## n x 1 matrix to a list conversion
#
M2L:=proc(M)
    convert(convert(M,Vector),list);
end proc:
#
## Highlit non-zero entries
#
spy:=proc(A)
    map(x->`if`(simplify(x)=0,0, `if`(simplify(x)=1,1,`*)) ,A);
end proc:
#
# Monomials of a set of polynomial in all indeterminates
#
PolyMonomials:=proc(S::list(ratpoly),Ord::{ShortTermOrder, TermOrder}) # Monomials
of a set of polynomials
local v,m,i,c,q;
    v:=indets(S);
    m:=[];
    for i from 1 to nops(S) do
        c:=[coeffs(expand(S[i]),v,'q')];
        m:=[op(m),q];
    end do;
    m:=MakeUnique(m);
    sort(m,(t1,t2)->testorder(t2,t1,Ord));
end proc:
#
## Monomias of a set of polynomials in given indeterminates
#
PolyVarsMonomials:=proc(S::list(ratpoly),Ord::{ShortTermOrder, TermOrder}) #
Monomials of a set of polynomials in variavbles of Ord
local v,m,i,c,q;
    v:={op(Ord)};
    m:=[];
    for i from 1 to nops(S) do
        c:=[coeffs(expand(S[i]),v,'q')];
        m:=[op(m),q];
    end do;
    m:=MakeUnique(m);
    sort(m,(t1,t2)->testorder(t2,t1,Ord));
end proc:

```

## 6-DOF Robot IK formulation

Given  $a_i$ ,  $d_i$ ,  $i = 1 \dots 6$ , and  $M_h$ , find parameters  $c_i$ ,  $s_i$ ,  $p_i$ ,  $r_i$  subject to

$$(1) \quad M_1 * M_2 * M_3 * M_4 * M_5 * M_6 = M_h$$

$$(2) (M11 * M12) * (M21 * M22) * (M31 * M32) * (M41 * M42) * (M51 * M52) * (M61 * M62) = Mh$$

$$(3) c_i^2 + s_i^2 = 1 \quad i = 1 \dots 6$$

$$(4) p_i^2 + r_i^2 = 1 \quad i = 1 \dots 6$$

[2] M. Raghavan, B. Roth. Kinematic Analysis of the 6R Manipulator of General Geometry. Int. Symposium on Robotic Research. pp. 264-269, Tokyo 1990.

Write (1) equivalently as

$$(5) M3 * M4 * M5 = M2^{-1} * M1^{-1} * Mh * M6^{-1}$$

$$(6) M31 * M32 * M41 * M42 * M51 * M52$$

$$= M22^{-1} * M21^{-1} * M12^{-1} * M11^{-1} * Mh * M62^{-1} * M61^{-1}$$

## Solution

### Symbolically from 12 equations to $Z p = 0$

The manipulator matrices

```
> M31 :=dhTs(3)[1]:
M32 :=dhTs(3)[2]:
M41 :=dhTs(4)[1]:
M42 :=dhTs(4)[2]:
M51 :=dhTs(5)[1]:
M52 :=dhTs(5)[2]:
iM22:=dhInvs(dhTs(2))[1]:
iM21:=dhInvs(dhTs(2))[2]:
iM12:=dhInvs(dhTs(1))[1]:
iM11:=dhInvs(dhTs(1))[2]:
Mh :=Matrix(4,4,[[lx,mx,nx,rx],[ly,my,ny,ry],[lz,mz,nz,rz],[0,0,0,1]]):
iM62:=dhInvs(dhTs(6))[1]:
iM61:=dhInvs(dhTs(6))[2]:
```

Let us first inspect the matrices.

```
> M31,M32,M41,M42,M51,M52,"=",iM22,iM21,iM12,iM11,Mh,iM62,iM61;
```

$$\begin{bmatrix} c3 & -s3 & 0 & 0 \\ s3 & c3 & 0 & 0 \\ 0 & 0 & 1 & d3 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} 1 & 0 & 0 & a3 \\ 0 & \lambda3 & -\mu3 & 0 \\ 0 & \mu3 & \lambda3 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} c4 & -s4 & 0 & 0 \\ s4 & c4 & 0 & 0 \\ 0 & 0 & 1 & d4 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} 1 & 0 & 0 & a4 \\ 0 & \lambda4 & -\mu4 & 0 \\ 0 & \mu4 & \lambda4 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} c5 & -s5 & 0 & 0 \\ s5 & c5 & 0 & 0 \\ 0 & 0 & 1 & d5 \\ 0 & 0 & 0 & 1 \end{bmatrix} \\ \begin{bmatrix} 1 & 0 & 0 & a5 \\ 0 & \lambda5 & -\mu5 & 0 \\ 0 & \mu5 & \lambda5 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} = \begin{bmatrix} 1 & 0 & 0 & -a2 \\ 0 & \lambda2 & \mu2 & 0 \\ 0 & -\mu2 & \lambda2 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} c2 & s2 & 0 & 0 \\ -s2 & c2 & 0 & 0 \\ 0 & 0 & 1 & -d2 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} 1 & 0 & 0 & -a1 \\ 0 & \lambda1 & \mu1 & 0 \\ 0 & -\mu1 & \lambda1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} c1 & s1 & 0 & 0 \\ -s1 & c1 & 0 & 0 \\ 0 & 0 & 1 & -d1 \\ 0 & 0 & 0 & 1 \end{bmatrix} \\ \begin{bmatrix} lx & mx & nx & rx \\ ly & my & ny & ry \\ lz & mz & nz & rz \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} 1 & 0 & 0 & -a6 \\ 0 & \lambda6 & \mu6 & 0 \\ 0 & -\mu6 & \lambda6 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} c6 & s6 & 0 & 0 \\ -s6 & c6 & 0 & 0 \\ 0 & 0 & 1 & -d6 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Notice that the the two last columns of iM61 are free of c6, s6 and so we can get six equations without the sixth variable.

Thus, take only those 6 equations to liminate c6, s6.

```
> M31,M32,M41,M42,M51,M52[1..4,3..4],"=",iM22,iM21,iM12,iM11,Mh,iM62,iM61[1..4,3..4];
```

$$\begin{bmatrix} c3 & -s3 & 0 & 0 \\ s3 & c3 & 0 & 0 \\ 0 & 0 & 1 & d3 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} 1 & 0 & 0 & a3 \\ 0 & \lambda3 & -\mu3 & 0 \\ 0 & \mu3 & \lambda3 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} c4 & -s4 & 0 & 0 \\ s4 & c4 & 0 & 0 \\ 0 & 0 & 1 & d4 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} 1 & 0 & 0 & a4 \\ 0 & \lambda4 & -\mu4 & 0 \\ 0 & \mu4 & \lambda4 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} c5 & -s5 & 0 & 0 \\ s5 & c5 & 0 & 0 \\ 0 & 0 & 1 & d5 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} 0 & a5 \\ -\mu5 & 0 \\ \lambda5 & 0 \\ 0 & 1 \end{bmatrix} \\ = \begin{bmatrix} 1 & 0 & 0 & -a2 \\ 0 & \lambda2 & \mu2 & 0 \\ 0 & -\mu2 & \lambda2 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} c2 & s2 & 0 & 0 \\ -s2 & c2 & 0 & 0 \\ 0 & 0 & 1 & -d2 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} 1 & 0 & 0 & -a1 \\ 0 & \lambda1 & \mu1 & 0 \\ 0 & -\mu1 & \lambda1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} c1 & s1 & 0 & 0 \\ -s1 & c1 & 0 & 0 \\ 0 & 0 & 1 & -d1 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} lx & mx & nx & rx \\ ly & my & ny & ry \\ lz & mz & nz & rz \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

$$\begin{bmatrix} 1 & 0 & 0 & -a6 \\ 0 & \lambda6 & \mu6 & 0 \\ 0 & -\mu6 & \lambda6 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} 0 & 0 \\ 0 & 0 \\ 1 & -d6 \\ 0 & 1 \end{bmatrix}$$

Do two following manipulations to "simplify" the set of equations:

1) Multiply both sides from the left by M22

```
> dhInv(im22),M31,M32,M41,M42,M51,M52[1..4,3..4], "=",im21,im12,im11,Mh,im62,im61[1..4,3..4];
```

$$\begin{bmatrix} 1 & 0 & 0 & a2 \\ 0 & \lambda2 & -\mu2 & 0 \\ 0 & \mu2 & \lambda2 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} c3 & -s3 & 0 & 0 \\ s3 & c3 & 0 & 0 \\ 0 & 0 & 1 & d3 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} 1 & 0 & 0 & a3 \\ 0 & \lambda3 & -\mu3 & 0 \\ 0 & \mu3 & \lambda3 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} c4 & -s4 & 0 & 0 \\ s4 & c4 & 0 & 0 \\ 0 & 0 & 1 & d4 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} 1 & 0 & 0 & a4 \\ 0 & \lambda4 & -\mu4 & 0 \\ 0 & \mu4 & \lambda4 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

$$\begin{bmatrix} c5 & -s5 & 0 & 0 \\ s5 & c5 & 0 & 0 \\ 0 & 0 & 1 & d5 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} 0 & a5 \\ -\mu5 & 0 \\ \lambda5 & 0 \\ 0 & 1 \end{bmatrix}, "=", \begin{bmatrix} c2 & s2 & 0 & 0 \\ -s2 & c2 & 0 & 0 \\ 0 & 0 & 1 & -d2 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} 1 & 0 & 0 & -a1 \\ 0 & \lambda1 & \mu1 & 0 \\ 0 & -\mu1 & \lambda1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} c1 & s1 & 0 & 0 \\ -s1 & c1 & 0 & 0 \\ 0 & 0 & 1 & -d1 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

$$\begin{bmatrix} lx & mx & nx & rx \\ ly & my & ny & ry \\ lz & mz & nz & rz \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} 1 & 0 & 0 & -a6 \\ 0 & \lambda6 & \mu6 & 0 \\ 0 & -\mu6 & \lambda6 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} 0 & 0 \\ 0 & 0 \\ 1 & -d6 \\ 0 & 1 \end{bmatrix}$$

2) Multiply both sides from the left by

```
> <<1,0,0,0>|<0,1,0,0>|<0,0,1,0>|<0,0,d2,1>>;
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & d2 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

```
> <<1,0,0,0>|<0,-1,0,0>|<0,0,1,0>|<0,0,d2,1>>.dhInv(im22),M31,M32,M41,M42,M51,M52[1..4,3..4],
```

```
"=",
```

```
<<1,0,0,0>|<0,-1,0,0>|<0,0,1,0>|<0,0,d2,1>>.im21,im12,im11,Mh,im62,im61[1..4,3..4];
```

$$\begin{bmatrix} 1 & 0 & 0 & a2 \\ 0 & -\lambda2 & \mu2 & 0 \\ 0 & \mu2 & \lambda2 & d2 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} c3 & -s3 & 0 & 0 \\ s3 & c3 & 0 & 0 \\ 0 & 0 & 1 & d3 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} 1 & 0 & 0 & a3 \\ 0 & \lambda3 & -\mu3 & 0 \\ 0 & \mu3 & \lambda3 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} c4 & -s4 & 0 & 0 \\ s4 & c4 & 0 & 0 \\ 0 & 0 & 1 & d4 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} 1 & 0 & 0 & a4 \\ 0 & \lambda4 & -\mu4 & 0 \\ 0 & \mu4 & \lambda4 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

$$\begin{bmatrix} c5 & -s5 & 0 & 0 \\ s5 & c5 & 0 & 0 \\ 0 & 0 & 1 & d5 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} 0 & a5 \\ -\mu5 & 0 \\ \lambda5 & 0 \\ 0 & 1 \end{bmatrix}, "=", \begin{bmatrix} c2 & s2 & 0 & 0 \\ s2 & -c2 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} 1 & 0 & 0 & -a1 \\ 0 & \lambda1 & \mu1 & 0 \\ 0 & -\mu1 & \lambda1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} c1 & s1 & 0 & 0 \\ -s1 & c1 & 0 & 0 \\ 0 & 0 & 1 & -d1 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

$$\begin{bmatrix} lx & mx & nx & rx \\ ly & my & ny & ry \\ lz & mz & nz & rz \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} 1 & 0 & 0 & -a6 \\ 0 & \lambda6 & \mu6 & 0 \\ 0 & -\mu6 & \lambda6 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} 0 & 0 \\ 0 & 0 \\ 1 & -d6 \\ 0 & 1 \end{bmatrix}$$

Do the multiplications, construct the equations, and take the 6 ones we are interested in.

Denote the left hand side by ee1 and the right hand side by ee2.

```
> ee1:=dhSimpl((<<1,0,0,0>|<0,-1,0,0>|<0,0,1,0>|<0,0,d2,1>>.dhInv(im22).M31.M32.M41.M42.M51.M52)[1..3,3..4],2):
```

```
ee2:=dhSimpl((<<1,0,0,0>|<0,-1,0,0>|<0,0,1,0>|<0,0,d2,1>>.im21.im12.im11.Mh.im62.im61)[1..3,3..4],2):
```

We have 6 equations in 10 unknowns s1, c1, s2, c2, s3, c3, s4, c4, s5, c5.

Let us generate more equations.

To do so, denote the columns of the left and right hand side as:

$$[l2 \ p2] = [l1 \ p1]$$

```
> l2:=ee1[1..3,1..1]:
```

```
p2:=ee1[1..3,2..2]:
```

```
l1:=ee2[1..3,1..1]:
```

```
p1:=ee2[1..3,2..2]:
```

```
> expand(p1[1,1]);
```

$$\begin{aligned}
& -d6 \mu6 mx c2 c1 + d6 \mu6 mx s2 \lambda1 s1 - d6 \mu6 my c2 s1 - d6 \mu6 my s2 \lambda1 c1 - d6 \mu6 s2 \mu1 mz - d6 \lambda6 nx c2 c1 \\
& + d6 \lambda6 nx s2 \lambda1 s1 - d6 \lambda6 ny c2 s1 - d6 \lambda6 ny s2 \lambda1 c1 - d6 \lambda6 s2 \mu1 nz - a6 lx c2 c1 + a6 lx s2 \lambda1 s1 \\
& - a6 ly c2 s1 - a6 ly s2 \lambda1 c1 - a6 s2 \mu1 lz + rx c2 c1 - rx s2 \lambda1 s1 + ry c2 s1 + ry s2 \lambda1 c1 + s2 \mu1 rz - s2 \mu1 dl \\
& - c2 a1
\end{aligned}$$

> **expand(p2[1,1]);**

$$\begin{aligned}
& a5 c5 c3 c4 - a5 c5 s3 \lambda3 s4 - a5 s5 \lambda4 c3 s4 - a5 s5 \lambda4 s3 \lambda3 c4 + a5 s5 s3 \mu3 \mu4 + d5 \mu4 c3 s4 + d5 \mu4 s3 \lambda3 c4 \\
& + d5 s3 \mu3 \lambda4 + a4 c3 c4 - a4 s3 \lambda3 s4 + s3 \mu3 d4 + c3 a3 + a2
\end{aligned}$$

New equations can be now generated by scalar and vector operations on the columns.

1)  $p2 \cdot p2 = p1 \cdot p1$  ... 1 new equation

2)  $p2 \cdot l2 = p1 \cdot l1$  ... 1 new equation

3)  $p2 \times l2 = p1 \times l1$  ... 3 new equations

4)  $(p1 \cdot p1) l1 - 2 (p1 \cdot l1) p1 = (p2 \cdot p2) l2 - 2 (p2 \cdot l2) p2$  ... 3 new equations

which can be derived from

$$A \times (B \times C) = (A \cdot C) B - (A \cdot B) C$$

using the substitution  $p = A = C$ ,  $l = B$

$$0 = A \times (B \times C) = (A \cdot C) B - (A \cdot B) C$$

and thus

$$-(A \cdot B) C = (A \cdot C) B - 2 (A \cdot B) C$$

we get

$$-(p1 \cdot l1) p1 = (p1 \cdot p1) l1 - 2 (p1 \cdot l1) p1$$

$$-(p2 \cdot l2) p2 = (p2 \cdot p2) l2 - 2 (p2 \cdot l2) p2$$

and using

$$p1 = p2 \quad \& \quad l1 = l2$$

we get

$$-(p1 \cdot l1) p1 = -(p2 \cdot l2) p2$$

$$(p1 \cdot p1) l1 - 2 (p1 \cdot l1) p1 = (p2 \cdot p2) l2 - 2 (p2 \cdot l2) p2$$

ad 1)  $p2 \cdot p2 = p1 \cdot p1$  ... 1 new equation

> **pp1:=MhSimpl(dhSimpl(dhSimpl(dhSimpl(Transpose(p1).p1,2),1),6));**  
**pp2:=dhSimpl(dhSimpl(dhSimpl(dhSimpl(Transpose(p2).p2,2),3),4),5);**

**pp1 :=**

$$\begin{aligned}
& [-2 rz dl + rx^2 + a6^2 + ry^2 + d6^2 + a1^2 + rz^2 - 2 rx c1 a1 - 2 ry s1 a1 + dl^2 + (2 a6 c1 a1 - 2 a6 rx) lx \\
& + (-2 a6 ry + 2 a6 s1 a1) ly + (-2 a6 rz + 2 a6 dl) lz + (-2 d6 \mu6 rx + 2 d6 \mu6 c1 a1) mx \\
& + (-2 d6 \mu6 ry + 2 d6 \mu6 s1 a1) my + (-2 d6 \mu6 rz + 2 d6 \mu6 dl) mz + (-2 d6 \lambda6 rx + 2 d6 \lambda6 c1 a1) nx \\
& + (2 d6 \lambda6 s1 a1 - 2 d6 \lambda6 ry) ny + (-2 d6 \lambda6 rz + 2 d6 \lambda6 dl) nz]
\end{aligned}$$

**pp2 :=**

$$\begin{aligned}
& [2 d2 \lambda2 d3 + 2 c3 a3 a2 + a2^2 + 2 a4 c3 c4 a2 + 2 d5 s3 \mu3 \lambda4 a2 + d2^2 + 2 d5 \mu4 c3 s4 a2 + a3^2 + 2 s3 \mu3 d4 a2 \\
& - 2 a4 s3 \lambda3 s4 a2 + d4^2 + 2 d2 \lambda2 a4 s4 \mu3 + 2 d2 \lambda2 d5 \lambda4 \lambda3 + 2 d2 \mu2 a4 s4 c3 \lambda3 - 2 d2 \mu2 d4 c3 \mu3 \\
& + 2 d2 \mu2 a4 s3 c4 + 2 d5 \lambda4 \lambda3 d3 + 2 d2 \lambda2 d4 \lambda3 + 2 d2 \mu2 s3 a3 + 2 d5 \mu4 s3 \lambda3 c4 a2 - 2 d2 \mu2 d5 \lambda4 c3 \mu3 \\
& + 2 d4 \lambda3 d3 + 2 d2 \mu2 d5 \mu4 s3 s4 + 2 d3 a4 s4 \mu3 + a4^2 + d5^2 + d3^2 + (-2 a5 s3 \lambda3 s4 a2 + 2 a5 c3 c4 a2
\end{aligned}$$

```

+ 2 d2 μ2 a5 s3 c4 + 2 a5 s4 μ3 d3 + 2 a5 a4 + 2 d2 λ2 a5 s4 μ3 + 2 d2 μ2 a5 s4 c3 λ3 + 2 a5 c4 a3) c5 + (
- 2 a5 λ4 s3 λ3 c4 a2 - 2 a5 λ4 c3 s4 a2 + 2 d2 λ2 a5 λ4 c4 μ3 - 2 d2 μ2 a5 λ4 s3 s4 + 2 a5 s3 μ3 μ4 a2
- 2 a5 λ4 s4 a3 + 2 d2 μ2 a5 λ4 c4 c3 λ3 - 2 d2 μ2 a5 μ4 c3 μ3 + 2 d4 a5 μ4 + 2 d3 a5 μ4 λ3 + 2 d2 λ2 a5 μ4 λ3
+ 2 a5 λ4 c4 μ3 d3) s5 + a52 + 2 d4 d5 λ4 - 2 d2 μ2 d5 μ4 c4 c3 λ3 + 2 d5 μ4 s4 a3 + 2 a4 c4 a3
- 2 d3 d5 μ4 c4 μ3 - 2 d2 λ2 d5 μ4 c4 μ3]
[ ad 2) p2 . l2 = p1 . l1 ... 1 new equation
[ > p11:=MhSimpl(dhSimpl(dhSimpl(dhSimpl(Transpose(p1).l1,2),1),6)):
p12:=dhSimpl(dhSimpl(dhSimpl(dhSimpl(Transpose(p2).l2,2),3),4),5)):
[ ad 3) p2 x l2 = p1 x l1 ... 3 new equations
[ > px11:=map(x->expand(x),convert(CrossProduct(convert(p1,Vector),convert(l1,Vector
)),Matrix)):
px12:=map(x->expand(x),convert(CrossProduct(convert(p2,Vector),convert(l2,Vector
)),Matrix)):
m1x:=map(x->expand(x),MhSimpl(dhSimpl(dhSimpl(dhSimpl(px11,2),1),6)):
m2x:=map(x->expand(x),dhSimpl(dhSimpl(dhSimpl(dhSimpl(px12,2),3),4),5)):
[ ad 4) (p1 . p1) l1 - 2 (p1 . l1) p1 = (p2 . p2) l2 - 2 (p2 . l2) p2 ... 3 new equations
[ > plp1:=map(x->expand(x),ScalarMultiply(l1,pp1[1,1]) -
ScalarMultiply(p1,2*pl1[1,1])):
plp2:=map(x->expand(x),ScalarMultiply(l2,pp2[1,1]) -
ScalarMultiply(p2,2*pl2[1,1])):
mp1:=MhSimpl(dhSimpl(dhSimpl(dhSimpl(simplify(plp1),2),1),6)):
mp2:=dhSimpl(dhSimpl(dhSimpl(dhSimpl(dhSimpl(simplify(plp2),2),3),4),5),1)):
[ Gather all 14 equations together:
[ > E1:=<p1,l1,pp1,p11,m1x,mp1>:
E2:=<p2,l2,pp2,p12,m2x,mp2>:
[ and construct its linear representation.
[ Let us look at c3, s3 as on parameters and consider the monomials of the 8 remaining
unknowns s1,c1,s2,c2,s4,c4,s5,c5 in the 6 equations.
[ 1) Notice that on the right hand side, there are the following monomials in s1,c1,s2,c2:
[ > t1:=<<s1*s2,s1*c2,c1*s2,c1*c2,s1,c1,s2,c2,1>>:
[ 2) Notice that on the left hand side, there are the following monomials in s4,c4,s5,c5:
[ > t2:=<<s4*s5,s4*c5,c4*s5,c4*c5,s4,c4,s5,c5,1>>:
[ Construct the linear representation in the above monomials:
[ > M1:=PolyCoeffMatrix(M2L(E1),M2L(t1),plex(op(indets(t1)))):
M2:=PolyCoeffMatrix(M2L(E2),M2L(t2),plex(op(indets(t2)))):
[ Check it.
[ > Transpose(simplify(E1-M1.t1));
Transpose(simplify(E2-M2.t2));
[
[0 0 0 0 0 0 0 0 0 0 0 0 0 0]
[0 0 0 0 0 0 0 0 0 0 0 0 0 0]
[ OK.
[ Move the constants from the right to the left and denote the left hand
side of the equations P and the right hand side of the equations Q.
[ > P:= <M2[1..14,1..8]|M2[1..14,9]-M1[1..14,9]>:
Q:= M1[1..14,1..8]:
[ Modify the corresponding monomial vectors and name them pp, qq.
[ > p:= t2:
q:= <t1[1..8,1]>:
[ We have 14 equations in 17 monomials of 10 unknowns constructed from 5 angles:
[ > Dimensions(P),`...`,Dimensions(p);
Dimensions(Q),`...`,Dimensions(q);
[
[14, 9, ..., 9, 1
[14, 8, ..., 8, 1
[ Check it.
[ > Transpose((-P.p+Q.q)-(M1.t1-M2.t2));
[
[0 0 0 0 0 0 0 0 0 0 0 0 0 0]
[ OK.
[ The matrices PP, QQ are semi-sparse:
[ > <Transpose(p),spy(P)>,<Transpose(q),spy(Q)>;

```





slightly missaligned 3-2-1 robot.

### Random general mechanism

```
> Mechanism := {
  a1=85, a2=280, a3=100, a4=0.1, a5=-0.1, a6=0.1,
  d1=350, d2=-0.1, d3=0.1, d4=315, d5=-0.1, d6=85,
  lambda1=cos(-Pi/2-Pi/100), mu1=sin(-Pi/2-Pi/100),
  lambda2=cos(Pi/110), mu2=sin(Pi/110),
  lambda3=cos(-Pi/2-Pi/90), mu3=sin(-Pi/2-Pi/90),
  lambda4=cos(Pi/2-Pi/120), mu4=sin(Pi/2-Pi/120),
  lambda5=cos(-Pi/2-Pi/95), mu5=sin(-Pi/2-Pi/95),
  lambda6=cos(-Pi/200), mu6=sin(-Pi/200)}:
```

Set randomly the joint angles and compute the corresponding Mh

```
> rndTh:=Pi*RandomMatrix(1,6)/200;
```

$$rndTh := \begin{bmatrix} \frac{\pi}{4} & \frac{\pi}{20} & -\frac{2\pi}{25} & -\frac{9\pi}{200} & -\frac{\pi}{4} & -\frac{11\pi}{100} \end{bmatrix}$$

```
> thetas :=
  map(x->x[1]=x[2],convert(<convert([theta1,theta2,theta3,theta4,theta5,theta6],Vector) |
    convert(simplify(rndTh),Vector)>,listlist)):
```

```
Position :=
  subs(thetas, {c1=cos(theta1), s1=sin(theta1), c2=cos(theta2), s2=sin(theta2), c3=cos(theta3), s3=sin(theta3),
```

```
  c4=cos(theta4), s4=sin(theta4), c5=cos(theta5), s5=sin(theta5), c6=cos(theta6), s6=sin(theta6)}):
```

```
MP := {op(Mechanism),op(Position)};
```

```
MhV := Matrix(4,4,[[lx,mx,nx,rx],[ly,my,ny,ry],[lz,mz,nz,rz],[0,0,0,1]]):
```

```
MhV :=
```

```
map(x->x[1]=x[2],convert(<convert(MhV[1..3,1..4],Vector) | convert(dhDKT(MP)[1..3,1..4],Vector)>,listlist)):
```

$$MP := \left\{ \mu_3 = -\sin\left(\frac{22\pi}{45}\right), c_4 = \cos\left(\frac{9\pi}{200}\right), c_2 = \cos\left(\frac{\pi}{20}\right), s_2 = \sin\left(\frac{\pi}{20}\right), c_1 = \frac{\sqrt{2}}{2}, s_1 = \frac{\sqrt{2}}{2}, \mu_6 = -\sin\left(\frac{\pi}{200}\right), \right.$$

$$c_6 = \cos\left(\frac{11\pi}{100}\right), a_1 = 85, a_2 = 280, a_3 = 100, a_4 = 0.1, a_5 = -0.1, a_6 = 0.1, d_1 = 350, d_2 = -0.1, d_3 = 0.1, d_4 = 315,$$

$$d_5 = -0.1, \lambda_2 = \cos\left(\frac{\pi}{110}\right), \mu_2 = \sin\left(\frac{\pi}{110}\right), \lambda_4 = \cos\left(\frac{59\pi}{120}\right), d_6 = 85, \mu_4 = \sin\left(\frac{59\pi}{120}\right), s_6 = -\sin\left(\frac{11\pi}{100}\right),$$

$$\lambda_5 = -\cos\left(\frac{93\pi}{190}\right), \mu_1 = -\sin\left(\frac{49\pi}{100}\right), \lambda_3 = -\cos\left(\frac{22\pi}{45}\right), s_4 = -\sin\left(\frac{9\pi}{200}\right), \mu_5 = -\sin\left(\frac{93\pi}{190}\right), c_5 = \frac{\sqrt{2}}{2}, s_5 = -\frac{\sqrt{2}}{2},$$

$$c_3 = \cos\left(\frac{2\pi}{25}\right), \lambda_1 = -\cos\left(\frac{49\pi}{100}\right), \lambda_6 = \cos\left(\frac{\pi}{200}\right), s_3 = -\sin\left(\frac{2\pi}{25}\right) \}$$

### Numerically from $Z p = 0$ to $f(x_3)=0$

Substitute the parameters of the manipulator and Mh into matrice P & Q and truncate by first converting it to Maple software floats (precision given by Digits) and then exactly to rational numbers

```
> MPh := {op(Mechanism),op(MhV)}:
  Qx := simplify(evalf(subs(MPh,Q))):
  Px := simplify(evalf(subs(MPh,P))):
  QxR := convert(Qx,rational,exact):
  PxR := convert(Px,rational,exact):
```

Look at the 8 largest singular values of QxR

```
> SingularValues(evalf(QxR));
```

```

378480.951865628073302112584649
378480.951865628073302112584647
304770.655569428774175857855054
83357.5565310114917490193163683
96.6051524679321534738938096784
96.6051524679321534738938092133
59.4337182736028838590927746217
59.4337182736028838590927744287
0.
0.
0.
0.
0.
0.

```

```
> svQxR:=evalf(SingularValues(evalf(QxR)))[1..8];
```

```

svQxR :=
378480.951865628073302112584649
378480.951865628073302112584647
304770.655569428774175857855054
83357.5565310114917490193163683
96.6051524679321534738938096784
96.6051524679321534738938092133
59.4337182736028838590927746217
59.4337182736028838590927744287

```

[ The condition number of QxR is not so bad

```
> svQxR[1]/svQxR[8];
```

```
6368.11834863322076468545706392
```

```
> QxRf:=evalf(QxR):
```

```
U, S, Vt := SingularValues(QxRf, output=['U', 'S', 'Vt']):
```

[ Use SVD to choose the best equations possible to continue:

Change the basis using the SVD of Q to get Z

$$P p = Q q$$

Make SVD of Q:  $Q = U S V^T$

$$P p = U S V^T q$$

Multiply by the inverse of U from the left. U has orthonormal columns  $\Rightarrow \text{inv}(U) = U^T$

$$U^T P p = S V^T q$$

Denote  $U^T P$  by A:  $A = U^T P$

$$A p = S V^T q$$

Split A into the first 8 rows, A8, and the rest 6 rows Z

$$\begin{bmatrix} [A8] p \\ [Z] \end{bmatrix} = \begin{bmatrix} [S8 V^T] q \\ [0] \end{bmatrix}$$

Get the equations from the last 6 rows

$$Z p = 0$$

```
> A := Transpose(U).PxR:
```

```
A8 := A[1..8,1..9]:
```

```
Z := A[9..14,1..9]:
```

[ Construct matrix Z

```
> `Z = ` , <Transpose(p),spy(Z)>;
```

$$Z = \begin{bmatrix} s_4 s_5 & s_4 c_5 & c_4 s_5 & c_4 c_5 & s_4 & c_4 & s_5 & c_5 & 1 \\ * & * & * & * & * & * & * & * & * \\ * & * & * & * & * & * & * & * & * \\ * & * & * & * & * & * & * & * & * \\ * & * & * & * & * & * & * & * & * \\ * & * & * & * & * & * & * & * & * \end{bmatrix}$$

[ We have 6 equations for 8 monomials but also recall that Z contains s3, c3:

[ > `indets(Z);`

{ c3, s3 }

[ > `evalf(Z[1..6,9]);`

```
[ -23.1439025483742743166575912865 c3 + 43.2737204114386719870953959470 s3
+ 55.2025229541276216445014219896 ]
[ 8.52865585879576648042176499926 c3 - 17.6254531697049390174587332137 s3
- 22.0351865097081846862772420950 ]
[ -30.6238402056738192926565560526 c3 + 97.1708255572899749764500381440 s3
+ 45.2983722865500019522611188548 ]
[ -0.270528288358414831025540111910 c3 + 0.857483717889941179625104494392 s3
- 1.17130256956595328358358210952 ]
[ 2.53874557335572477551923859760 c3 - 3.06931966798848983398459559096 s3
- 4.20335467900547641539041211177 ]
[ 43.8455576056752250059309786716 c3 - 138.216889316769289751762781980 s3
+ 457.521299735815002292038143946 ]
```

[ We will now proceed by choosing a different circle parameterization.

$$x/1 = s/(1+c)$$

$$s = x(1+c)$$

$$s^2 + c^2 = 1$$

$$x^2(1+c)^2 + c^2 = 1$$

$$x^2 + 2x^2c + x^2c^2 + c^2 - 1 = 0$$


$$(1+x^2)c^2 + 2x^2c + x^2 - 1 = 0$$

$$c = \frac{-2x^2 + \sqrt{4x^4 - 4(1+x^2)(x^2-1)}}{2(1+x^2)}$$

$$c = \frac{-2x^2 + 2\sqrt{1}}{2(1+x^2)}$$

$$c = \frac{-x^2 + 1}{1+x^2}$$

$$s = x(1+c) = \frac{2x}{1+x^2}$$



which then gives

$$s_i = \frac{2x}{1+x^2}$$

$$c_i = \frac{1-x^2}{1+x^2}$$

for  $i = 4, 5$

[ > `x:=subs([s4=(2*x4)/(1+x4^2),c4=(1-x4^2)/(1+x4^2),s5=(2*x5)/(1+x5^2),c5=(1-x5^2)/(1+x5^2)],p);`  
`Transpose(x);`







```
[-0.126849485579164462396458822450
-0.126329378446108174788185342459
-2.15855009557149911023027128510
-2.15818402406524636065874729650]
```

```
> map(x->subs(x3=x,evalf([(2*x3)/(1+x3^2),(1-x3^2)/(1+x3^2)]),x3s);
evalf(subs(Position,[s3,c3]));
```

```
[ -0.249681399691693709152980831236, 0.968328042890422900593369448699 ]
[ -0.248689887164854788246354413198, 0.968583161128631119489123205768 ]
[ -0.762827701440832719790586335390, -0.646601807849696456815517446505 ]
[ -0.762911356971281781987225129082, -0.646503102393358558234682800357 ]
[ -0.248689887164854788242283746006, 0.968583161128631119490168375465 ]
```



**Substitute back and solve for all angles**

[ Solve for x5, x5 from Mx3

```
> ix5:=1;
```

```
ix5 := 1
```

```
> Mx3s:=subs(x3=x3s[ix5],Mx3):
```

```
`Mx3s = ` , <Transpose(<myy>),spy(Mx3s)>;
```

```
Mx3s = ,
```

$x^4^3$	$x^5$	$x^5$	$x^4^3$	$x^4^3$	$x^4^2$	$x^5$	$x^4^2$	$x^4$	$x^5$	$x^5$	$x^5$	1
*	*	*	*	*	*	*	*	*	*	0	0	0
*	*	*	*	*	*	*	*	*	*	0	0	0
*	*	*	*	*	*	*	*	*	*	0	0	0
*	*	*	*	*	*	*	*	*	*	0	0	0
*	*	*	*	*	*	*	*	*	*	0	0	0
0	0	0	*	*	*	*	*	*	*	*	*	*
0	0	0	*	*	*	*	*	*	*	*	*	*
0	0	0	*	*	*	*	*	*	*	*	*	*
0	0	0	*	*	*	*	*	*	*	*	*	*
0	0	0	*	*	*	*	*	*	*	*	*	*
0	0	0	*	*	*	*	*	*	*	*	*	*

[ Notice that the matrix has numerical dimension 11. The last singular value is much smaller than the last but one singular value.

```
> sv,V:= SingularValues(Mx3s, output=['S','Vt']):
sv/sv[1];
```

```
[ 1.00000000000000000000000000000000
0.995483526527741653381120290172
0.607286577005152478482407052209
0.606321547351437105015324368999
0.422554074632193240775016153098
0.409891196321783341966423577500
0.120309017261210028495347337441 10^-5
0.119784981231630056580361372018 10^-5
0.153009111291628914224812419837 10^-6
0.119729414552328288859604733388 10^-6
0.422445051843191146987157657637 10^-7
0.205169085964551981131966753823 10^-24]
```

[ Therefore, the approximate solution of Mx3 X = 0 is the singular vector corresponding the the smallest singular value.

```
> myys := Transpose(V[12..12,1..12])/V[12,12]:
Norm(Transpose(Mx3s.(myys/myys[12,1])));
```

```
0.132111243637079780 10^-12
```

```
> myys[12,1];
```

```
0.99999999999999999999999999999999
```

[ We see Mx3s maps myys very close to the zero vector.

[ By comparing

```
> <<myy>|<myys>>;
```

$x^4^3 x^5^2$	-0.247636982319954133594972036079 10 <sup>8</sup>
$x^5 x^4^3$	-0.598159659037719504871426381242 10 <sup>8</sup>
$x^4^3$	-0.144483660860411959152258064267 10 <sup>9</sup>
$x^4^2 x^5^2$	47192.8194324290076505834466265
$x^5 x^4^2$	113992.831427165111080697049113
$x^4^2$	275346.244896164928642903732631
$x^4 x^5^2$	-89.9365751075127425810625963961
$x^4 x^5$	-217.239083586592365667078024315
$x^4$	-524.734451819127464432911394766
$x^5^2$	0.171394454488051929052280907877
$x^5$	0.413998133430478675856545082910
1	0.99999999999999999999999999999999

[ we see that

```
> x4s := myys[9,1]/myys[12,1];
x5s := myys[11,1]/myys[12,1];
```

```
x4s := -524.734451819127464432911394767
x5s := 0.413998133430478675856545082910
```

[ Compute sines and cosines from x3s, x3s and x5s:

```
> s3s := subs(x3=x3s[ixs],evalf((2*x3)/(1+x3^2)));
c3s := subs(x3=x3s[ixs],evalf((1-x3^2)/(1+x3^2)));
s4s := subs(x4=x4s,evalf((2*x4)/(1+x4^2)));
c4s := subs(x4=x4s,evalf((1-x4^2)/(1+x4^2)));
s5s := subs(x5=x5s,evalf((2*x5)/(1+x5^2)));
c5s := subs(x5=x5s,evalf((1-x5^2)/(1+x5^2)));
```

```
s3s := -0.249681399691693709152980831236
c3s := 0.968328042890422900593369448699
s4s := -0.00381143782252335762955430783214
c4s := -0.999992736444482899828095974664
s5s := 0.706846667825268542321093114904
c5s := 0.707366798898785235656403192594
```

[ Check trigonometric identities:

```
> [c3s^2+s3s^2-1,c4s^2+s4s^2-1,c5s^2+s5s^2-1];
```

```
[-0.8 10-29, 0., -0.4 10-29]
```

[ Check the solutions

```
> [s3s,c3s]-evalf(subs(Position,[s3,c3]));
[s4s,c4s]-evalf(subs(Position,[s4,c4]));
[s5s,c5s]-evalf(subs(Position,[s5,c5]));
```

```
[-0.000991512526838920910697085230, -0.000255118238208218896798926766]
[0.137089794115059303535784812725, -1.99001639416104046707953169114]
[1.41395344901181606672193747701, 0.000260017712237711255558830489]
```

[ Store the solution in a form convenient for subst

```
> S345:={s3=s3s,c3=c3s,s4=s4s,c4=c4s,s5=s5s,c5=c5s};
```

[ Substitute to the original equations

$$[A8] p = [S8 V^T] q$$

$$q = V \text{inv}(S8) A8 p$$

```
> qs:=Transpose(Vt).DiagonalMatrix(map(s->1/s,S[1..8])).subs(S345,A8.p):
> `<q|qs>`,` `<q|qs>;
```



$$\langle q|qs \rangle = \begin{bmatrix} s1 & s2 & 0.111018112225788286065344088209 \\ s1 & c2 & 0.698058000436117533671343235303 \\ c1 & s2 & 0.111104741712675087501428458111 \\ c1 & c2 & 0.698602707438703416253676671768 \\ s1 & & 0.706830951025433534814876548608 \\ c1 & & 0.707382503793074127329197842419 \\ s2 & & 0.157064588439997918448488385980 \\ c2 & & 0.987588332760851465844772713835 \end{bmatrix}$$

[ By comparison, we see that we get the solutions for s1, c1 and s2, c2 as

```
> s1s:=qs[5,1];
c1s:=qs[6,1];
s2s:=qs[7,1];
c2s:=qs[8,1];

s1s := 0.706830951025433534814876548608
c1s := 0.707382503793074127329197842419
s2s := 0.157064588439997918448488385980
c2s := 0.987588332760851465844772713835
```

[ which are correct as shows

```
> [s1s,c1s]-evalf(subs(Position,[s1,c1]));
[s2s,c2s]-evalf(subs(Position,[s2,c2]));

[-0.000275830161113989585967813497, 0.000275722606526602928353480314]
[0.000630123399767049438383066513, -0.000100007834286260345267533858]
```

[ Store the solutions

```
> S12:={s1=s1s,c1=c1s,s2=s2s,c2=c2s}:
```

[ Finally, let's get the solution to s6, c6. Let's return to the original, matrix equation and let's take the two left columns of the matrices:

```
> e61:=dhSimpl((<<1,0,0,0>|<0,-1,0,0>|<0,0,1,0>|<0,0,d2,1>>.dhInv(im22).M31.M32.M4
1.M42.M51.M52)[1..3,1..2],2):
e62:=dhSimpl((<<1,0,0,0>|<0,-1,0,0>|<0,0,1,0>|<0,0,d2,1>>.im21.im12.im11.Mh.im62
.im61)[1..3,1..2],2):
```

[ Substitute the known solutions

```
> e61v:=subs(S12,subs(S345,evalf(subs(MPh,e61)))));
e62v:=subs(S12,subs(S345,evalf(subs(MPh,e62)))));

e61v := [-0.508386605780919387456065424534 -0.0226061239021880614183240818139]
[-0.860869969380662968124569627010 0.0378567698279086004139766922711]
[0.0211176438338292317319590480029 0.999027441134784680871132509606]
```

```
e62v :=
[-0.257206945998259271163405364905 s6 + 0.439104275244625085662217964516 c6 ,
0.257206945998259271163405364905 c6 + 0.439104275244625085662217964516 s6]
[0.779076839793321179600158212797 c6 - 0.368197660252756499385775835103 s6 ,
0.779076839793321179600158212797 s6 + 0.368197660252756499385775835103 c6]
[0.447466996679469745494628105975 c6 + 0.893462405422890633380972151302 s6 ,
0.447466996679469745494628105975 s6 - 0.893462405422890633380972151302 c6]
```

[ and form the equations

```
> e6:=convert(map(p->sort(p),e62v-e61v),Vector);

e6 :=
[0.439104275244625085662217964516 c6 - 0.257206945998259271163405364905 s6
+ 0.508386605780919387456065424534]
[0.779076839793321179600158212797 c6 - 0.368197660252756499385775835103 s6
+ 0.860869969380662968124569627010]
[0.447466996679469745494628105975 c6 + 0.893462405422890633380972151302 s6
- 0.0211176438338292317319590480029]
[0.257206945998259271163405364905 c6 + 0.439104275244625085662217964516 s6
+ 0.0226061239021880614183240818139]
[0.368197660252756499385775835103 c6 + 0.779076839793321179600158212797 s6
- 0.0378567698279086004139766922711]
```

```
[ -0.893462405422890633380972151302 c6 + 0.447466996679469745494628105975 s6  
- 0.999027441134784680871132509606 ]
```

```
[ Convert them into the matrix form
```

```
> t6:=[c6,s6,1]:  
M6:=PolyCoeffMatrix(M2L(e6),t6):  
<Transpose(<t6>),M6>;
```

```
[ c6 , s6 , 1 ]
```

```
[ 0.439104275244625085662217964516 , -0.257206945998259271163405364905 ,  
0.508386605780919387456065424534 ]  
[ 0.779076839793321179600158212797 , -0.368197660252756499385775835103 ,  
0.860869969380662968124569627010 ]  
[ 0.447466996679469745494628105975 , 0.893462405422890633380972151302 ,  
-0.0211176438338292317319590480029 ]  
[ 0.257206945998259271163405364905 , 0.439104275244625085662217964516 ,  
0.0226061239021880614183240818139 ]  
[ 0.368197660252756499385775835103 , 0.779076839793321179600158212797 ,  
-0.0378567698279086004139766922711 ]  
[ -0.893462405422890633380972151302 , 0.447466996679469745494628105975 ,  
-0.999027441134784680871132509606 ]
```

```
[ and choose the first two to get the solution:
```

```
> cs6s:=MatrixInverse(-M6[1..2,1..2]).M6[1..2,3..3]:  
c6s:=cs6s[1,1];  
s6s:=cs6s[2,1];
```

```
c6s := -0.88446913664344970919923066668
```

```
s6s := 0.46659869970879539502749362701
```

```
[ and compare them with the ground truth
```

```
> [s6s,c6s]-evalf(subs(Position,[s6,c6]));  
[ 0.805336619954086776249777981977, -1.82534990559767518152334908578 ]
```

```
[ Store the solution
```

```
> S6:={s6=cs6s[2,1],c6=cs6s[1,1]};
```

```
S6 := { c6 = -0.88446913664344970919923066668, s6 = 0.46659869970879539502749362701 }
```

```
>
```